unittest
{
	assert( [ "axyb", "abyxb" ].parse.expand.solve == "axb" );
	assert( [ "aa", "xayaz" ].parse.expand.solve == "aa" );
	assert( [ "a", "z" ].parse.expand.solve == "" );
	assert( [ "abracadabra", "avadakedavra" ].parse.expand.solve == "aaadara" );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto s = input.front;
	input.popFront;
	auto t = input.front;
	return tuple( s, t );
}

auto solve( string s, string t )
{
	// dp[ si ][ ti ] = s[ 0 .. si ]とt[ 0 .. ti ]の最長共通部分列の長さ
	auto dp = new long[][]( 1 + s.length, 1 + t.length );
	for( auto si = 0L; si < s.length; si++ )
	{
		for( auto ti = 0L; ti < t.length; ti++ )
		{
			if( s[ si ] == t[ ti ] )
			{
				// 文字が一致する場合、長さを+1する
				dp[ si + 1 ][ ti + 1 ] = dp[ si ][ ti ] + 1;
			}
			else
			{
				// 文字が一致しない場合、一つ前の長いほうを使う
				dp[ si + 1 ][ ti + 1 ] = max( dp[ si ][ ti + 1 ], dp[ si + 1 ][ ti ] );
			}
		}
	}
	
	// dpを逆に辿って共通部分列を復元
	auto ans = "";
	auto si = s.length;
	auto ti = t.length;
	while( 0 < si && 0 < ti )
	{
		if( dp[ si ][ ti ] == dp[ si - 1 ][ ti ] )
		{
			si--;
			continue;
		}
		
		if( dp[ si ][ ti ] == dp[ si ][ ti - 1 ] )
		{
			ti--;
			continue;
		}
		
		si--;
		ti--;
		ans = s[ si ] ~ ans;
	}
	
	return ans;
}
